On the Perimeter of an Ellipse

Paul Abbott

School of Physics, M013
The University of Western Australia
35 Stirling Highway
Crawley WA 6009, Australia

~~paul@physics.uwa.edu.au~~
~~physics.uwa.edu.au/~paul~~

Computing accurate approximations to the perimeter of an ellipse is a favourite problem of amateur mathematicians, even attracting luminaries such as Ramanujan [1, 2, 3]. As is well known, the perimeter, $\mathcal{P}$, of an ellipse with semimajor axis a and semiminor axis b can be expressed exactly as a complete elliptic integral of the second kind, which can also be written as a Gaussian hypergeometric function,

$$\mathcal{P} = 4\,a\,E\left(1-\frac{b^2}{a^2}\right) = 2\,\pi\,a\;{}_2F_1\left(\frac{1}{2}, -\frac{1}{2}; 1; 1-\frac{b^2}{a^2}\right). \tag{1}$$

What is less well known is that the various exact forms attributed to Maclaurin, Gauss−Kummer, and Euler, are related via quadratic transformation formulae for hypergeometric functions. In this way we obtain additional identities, including a particularly elegant formula, symmetric in a and b,

$$\mathcal{P} = 2\,\pi\sqrt{a\,b}\;P_{\frac{1}{2}}\left(\frac{a^2+b^2}{2\,a\,b}\right), \tag{2}$$

where $P_\nu(z)$ is a Legendre function.

Approximate formulae can be obtained by truncating the series representations of exact formulas. For example, Kepler used the geometric mean, $\mathcal{P} \approx 2\,\pi\sqrt{a\,b}$. In this paper, we examine the properties of a number of approximate formulas, using series methods, polynomial interpolation, rational polynomial approximants, and minimax methods.

■ Cartesian Equation

The Cartesian equation for an ellipse with centre at (0, 0), semimajor axis a, and semiminor axis b reads

$$\mathcal{E}(x\_, y\_) = \left(\frac{x}{a}\right)^2 + \left(\frac{y}{b}\right)^2 == 1;$$

Introducing the parameter ϕ into the Cartesian coordinates, as $(x = a\sin(\phi),\ y = b\cos(\phi))$, one verifies that the ellipse equation is satisfied.

Simplify[$\mathcal{E}(a\sin(\phi),\ b\cos(\phi))$]

True

■ Arclength

In general, the parametric arclength is defined by

$$\mathcal{L} = \int_{\phi_1}^{\phi_2} \sqrt{\left(\frac{\partial x}{\partial \phi}\right)^2 + \left(\frac{\partial y}{\partial \phi}\right)^2}\, d\phi \tag{3}$$

The arclength of an ellipse as a function of the parameter ϕ is an (incomplete) elliptic integral of the second kind.

$$\mathcal{L}(\phi\_) = \text{With}\Big[\{x = a\sin(\phi),\ y = b\cos(\phi)\},\ \text{Simplify}\Big[\int \sqrt{\left(\frac{\partial x}{\partial \phi}\right)^2 + \left(\frac{\partial y}{\partial \phi}\right)^2}\, d\phi,\ a > b > 0 \bigwedge 0 < \phi < \frac{\pi}{2}\Big]\Big]$$

$$a\, E\left(\phi \,\middle|\, 1 - \frac{b^2}{a^2}\right)$$

Since,

$$\mathcal{L}(0) == 0$$

True

the arclenth of the ellipse is

$$\mathcal{L}(\phi) = a\, E(\phi \mid e^2) \tag{4}$$

where the eccentricity, e, is defined by

$$e(\text{a\_},\ \text{b\_}) = \sqrt{1 - \frac{b^2}{a^2}};$$

■ Perimeter

Since the parameter ranges over $0 \le \phi \le \pi/2$ for one quarter of the ellipse, the perimeter of the ellipse is

$$\mathcal{P}_1(\text{a\_},\ \text{b\_}) = 4\,\mathcal{L}\left(\frac{\pi}{2}\right)$$

$$4\, a\, E\left(1 - \frac{b^2}{a^2}\right)$$

That is $\mathcal{P} = 4\, a\, E(e^2)$ where $E(m)$ is the complete elliptic integral of the second kind.

□ Alternative Expressions for the Perimeter

The above expression for the perimeter of the ellipse is *unsymmetrical* with respect to the parameters a and b. This is “unphysical” in that both parameters, being lengths of the (major and minor) axes, should be on the same footing. We can expect that a *symmetric* formula, when truncated, will more accurately approximate the perimeter for both $a \ge b$ and $a \le b$.

Noting that the complete elliptic integral is a gaussian hypergeometric function,

$${}_2F_1\left(\frac{1}{2}, -\frac{1}{2}; 1; z\right)$$

$$\frac{2\,E(z)}{\pi}$$

one obtains Maclaurin's 1742 formula (see [2])

$$\mathcal{P}_1(a, b) = 2\pi a\, {}_2F_1\left(\frac{1}{2}, -\frac{1}{2}; 1; e(a, b)^2\right)$$

True

Equivalent alternative expressions for the perimeter of the ellipse can be obtained from quadratic transformation formulæ for gaussian hypergeometric functions. For example, using functions.wolfram.com/07.23.17.0106.01,

$$\text{Simplify}\left[{}_2F_1(\alpha, \beta; 2\beta; z) = \frac{{}_2F_1\left(\alpha, \alpha-\beta+\frac{1}{2}; \beta+\frac{1}{2}; \left(\frac{1-\sqrt{1-z}}{\sqrt{1-z}+1}\right)^2\right)}{\left(\frac{1}{2}\left(\sqrt{1-z}+1\right)\right)^{2\alpha}} \,/.\right.$$

$$\left.\left\{\beta \to \frac{1}{2}, \alpha \to -\frac{1}{2}, z \to e(a, b)^2\right\}, a > b > 0\right]$$

$$4\,a\,E\left(1-\frac{b^2}{a^2}\right) = (a+b)\,\pi\, {}_2F_1\left(-\frac{1}{2}, -\frac{1}{2}; 1; \frac{(a-b)^2}{(a+b)^2}\right)$$

and noting that

$$\frac{(a-b)^2}{(a+b)^2} = 1 - \frac{4\,a\,b}{(a+b)^2} \text{ // Simplify}$$

True

one obtains the following symmetric formula

$$\mathcal{P}_2(\text{a\_}, \text{b\_}) = \pi(a+b)\, {}_2F_1\left(-\frac{1}{2}, -\frac{1}{2}; 1; 1-\frac{4\,a\,b}{(a+b)^2}\right);$$

first obtained by Ivory (1796), but known as the Gauss–Kummer series (see [2]).

Introducing the homogenous symmetric parameter $h = \frac{(a-b)^2}{(a+b)^2} = 1 - \frac{4\,a\,b}{(a+b)^2}$, one has (*c.f.* mathworld.wolfram.com/Ellipse.html),

$$\pi(a+b)\, {}_2F_1\left(-\frac{1}{2}, -\frac{1}{2}; 1; h\right) \text{ // FunctionExpand // Simplify}$$

$$2\,(a+b)\,(2\,E(h) + (h-1)\,K(h))$$

Explicitly, the Gauss–Kummer series reads

$$\mathcal{P}_3(\text{a\_}, \text{b\_}) = \text{FullSimplify}[\mathcal{P}_2(a, b) \text{ // FunctionExpand}, a > b > 0]$$

$$4\,(a+b)\,E\left(1-\frac{4\,a\,b}{(a+b)^2}\right) - \frac{8\,a\,b\,K\left(1-\frac{4\,a\,b}{(a+b)^2}\right)}{a+b}$$

Instead, using functions.wolfram.com/07.23.17.0103.01, one obtains Euler's 1773 formula (see also [2]):

$$ {}_2F_1(\alpha, \beta; 2\beta; z) = \frac{{}_2F_1\left(\frac{\alpha}{2}, \frac{\alpha+1}{2}; \beta+\frac{1}{2}; \frac{z^2}{(2-z)^2}\right)}{\left(1-\frac{z}{2}\right)^{\alpha}} \,/. $$

$$ \left\{\beta \to \frac{1}{2}, \alpha \to -\frac{1}{2}, z \to e(a, b)^2\right\} \text{ // Simplify} $$

$$ 4\,E\left(1-\frac{b^2}{a^2}\right) = \sqrt{\frac{2\,b^2}{a^2}+2}\;\pi\,{}_2F_1\left(-\frac{1}{4}, \frac{1}{4}; 1; \frac{(a^2-b^2)^2}{(a^2+b^2)^2}\right) $$

The hidden symmetry with respect to the interchange $a \leftrightarrow b$ is revealed.

FullSimplify[%, $b > a > 0$]

$$ b\,E\left(1-\frac{a^2}{b^2}\right) = a\,E\left(1-\frac{b^2}{a^2}\right) $$

Defining

$$ \mathcal{P}_4(\text{a\_}, \text{b\_}) = \pi\sqrt{2\,(a^2+b^2)}\;{}_2F_1\left(\frac{1}{4}, -\frac{1}{4}; 1; \left(\frac{a^2-b^2}{a^2+b^2}\right)^2\right); $$

one can directly check the formula.

Simplify[$\mathcal{P}_4(a, b) == \mathcal{P}_1(a, b)$ // FunctionExpand, $a > b > 0$]

True

□ Other identities

There are many other possible transformation formulas that can be applied to obtain alternative expressions for the perimeter. For example, using functions.wolfram.com/07.23.17.0054.01 one obtains the following formula,

$$ \mathcal{P}_5(\text{a\_}, \text{b\_}) = \mathcal{P}_2(a, b) \,/.\, {}_2F_1(\text{a\_}, \text{b\_}; \text{c\_}; \text{z\_}) \to (1-z)^{-a-b+c}\;{}_2F_1(c-a, c-b; c; z) $$

$$ \frac{16\,a^2\,b^2\,\pi\,{}_2F_1\left(\frac{3}{2}, \frac{3}{2}; 1; 1-\frac{4\,a\,b}{(a+b)^2}\right)}{(a+b)^3} $$

The perimeter can also be expressed in terms of Legendre functions (see sections 8.13 and 15.4 of [4]). For example, using 15.4.15 of [4] one obtains an elegant and simple symmetric formula

$$ \mathcal{P}_6(\text{a\_}, \text{b\_}) = \text{Simplify}\Big[\mathcal{P}_2(a, b) \,/.\, {}_2F_1(\text{a\_}, \text{b\_}; \text{c\_}; \text{x\_}) :\to \Gamma(a-b+1)\,(1-x)^{-b}\,(-x)^{\frac{b-a}{2}}\,P_{-b}^{b-a}\left(\frac{1+x}{1-x}\right) /; c == a-b+1, a > 0 \wedge b > 0\Big] $$

$$ 2\sqrt{a\,b}\;\pi\,P_{\frac{1}{2}}\left(\frac{a^2+b^2}{2\,a\,b}\right) $$

This form can be used to prove that the perimeter of an ellipse is a homogenous mean (*c.f.* [5]), extending the arithmetic−geometric mean (AGM) already used as a tool for computing elliptic integrals [6].

Using functions.wolfram.com/07.07.26.0001.01, this gives yet another formula involving complete elliptic integrals.

$\mathcal{P}_7$ (a_, b_) =

$\mathcal{P}_6$ (a, b) /. $P_{\nu\_}$ (z_) → $_2F_1\left(-\nu, \nu+1; 1; \frac{1-z}{2}\right)$ // FunctionExpand // Simplify

$$4\sqrt{a\,b}\left(2\,E\left(-\frac{(a-b)^2}{4\,a\,b}\right)-K\left(-\frac{(a-b)^2}{4\,a\,b}\right)\right)$$

□ Comparisons

Here we compare the seven formulas obtained above for $b = 2\,a$,

**Simplify[{$\mathcal{P}_1$ (a, 2 a), $\mathcal{P}_2$ (a, 2 a), $\mathcal{P}_3$ (a, 2 a),
$\mathcal{P}_4$ (a, 2 a), $\mathcal{P}_5$ (a, 2 a), $\mathcal{P}_6$ (a, 2 a), $\mathcal{P}_7$ (a, 2 a)}, a > 0]**

$$\left\{4\,a\,E(-3),\ 3\,a\,\pi\,{}_2F_1\left(-\frac{1}{2}, -\frac{1}{2}; 1; \frac{1}{9}\right),\ \frac{4}{3}\,a\left(9\,E\left(\frac{1}{9}\right)-4\,K\left(\frac{1}{9}\right)\right),\right.$$
$$\sqrt{10}\,a\,\pi\,{}_2F_1\left(\frac{1}{4}, -\frac{1}{4}; 1; \frac{9}{25}\right),\ \frac{64}{27}\,a\,\pi\,{}_2F_1\left(\frac{3}{2}, \frac{3}{2}; 1; \frac{1}{9}\right),$$
$$\left.2\sqrt{2}\,a\,\pi\,P_{\frac{1}{2}}\left(\frac{5}{4}\right),\ 4\sqrt{2}\,a\left(2\,E\left(-\frac{1}{8}\right)-K\left(-\frac{1}{8}\right)\right)\right\}$$

***N*[%]**

{9.68845 a, 9.68845 a, 9.68845 a, 9.68845 a, 9.68845 a, 9.68845 a, 9.68845 a}

Equal @@ %

True

and for $b = a/3$.

**Simplify[{$\mathcal{P}_1\left(a, \frac{a}{3}\right)$, $\mathcal{P}_2\left(a, \frac{a}{3}\right)$, $\mathcal{P}_3\left(a, \frac{a}{3}\right)$,
$\mathcal{P}_4\left(a, \frac{a}{3}\right)$, $\mathcal{P}_5\left(a, \frac{a}{3}\right)$, $\mathcal{P}_6\left(a, \frac{a}{3}\right)$, $\mathcal{P}_7\left(a, \frac{a}{3}\right)$}, a > 0]**

$$\left\{4\,a\,E\left(\frac{8}{9}\right),\ \frac{4}{3}\,a\,\pi\,{}_2F_1\left(-\frac{1}{2}, -\frac{1}{2}; 1; \frac{1}{4}\right),\right.$$
$$\frac{2}{3}\,a\left(8\,E\left(\frac{1}{4}\right)-3\,K\left(\frac{1}{4}\right)\right),\ \frac{2}{3}\sqrt{5}\,a\,\pi\,{}_2F_1\left(\frac{1}{4}, -\frac{1}{4}; 1; \frac{16}{25}\right),$$
$$\left.\frac{3}{4}\,a\,\pi\,{}_2F_1\left(\frac{3}{2}, \frac{3}{2}; 1; \frac{1}{4}\right),\ \frac{2\,a\,\pi\,P_{\frac{1}{2}}\left(\frac{5}{3}\right)}{\sqrt{3}},\ \frac{a\left(8\,E\left(-\frac{1}{3}\right)-4\,K\left(-\frac{1}{3}\right)\right)}{\sqrt{3}}\right\}$$

***N*[%]**

{4.45496 a, 4.45496 a, 4.45496 a, 4.45496 a, 4.45496 a, 4.45496 a, 4.45496 a}

Equal @@ %

True

■ Numerical Approximation

At www.ebyte.it/library/docs/math05a/EllipsePerimeterApprox05.html [1]one is encouraged to search for “an efficient formula using only the four algebraic operations (if possible, avoiding even square−root) with a maximum error below 10 parts per million. If would be also nice if such a formula were exact for both the circle and the degenerate flat ellipse.”

The Gauss−Kummer series expressed as a function of the homogenous variable $h = 1 - 4\,a\,b\big/(a+b)^2$, reads

$$\textbf{GaussKummer[h\_]} = \frac{\mathcal{P}_2(a, b)}{a+b} \;/.\; (a+b) \to 2\sqrt{a\,b} \,\Big/ \sqrt{1-h}$$

$$\pi\,{}_2F_1\left(-\frac{1}{2}, -\frac{1}{2}; 1; h\right)$$

Series expansions

The series expansion about $h = 0$ is useful for small h.

$$\textbf{GaussKummer}[h] + O[h]^9$$

$$\pi + \frac{\pi h}{4} + \frac{\pi h^2}{64} + \frac{\pi h^3}{256} + \frac{25\pi h^4}{16384} + \frac{49\pi h^5}{65536} + \frac{441\pi h^6}{1048576} + \frac{1089\pi h^7}{4194304} + \frac{184041\pi h^8}{1073741824} + O(h^9)$$

Around $h = 1$, terms in $\log(1-h)$ arise.

$$\textbf{Simplify[Series[GaussKummer}[h], \{h, 1, 2\}], 0 < h < 1]$$

$$4 + (h-1) + \frac{1}{16}\left(-2\log(1-h) - 4\psi^{(0)}\left(\frac{3}{2}\right) - 4\gamma + 3\right)(h-1)^2 + O\left((h-1)^3\right)$$

Using functions.wolfram.com/07.23.06.0015.01 we can obtain the general term of this series (*c.f.* 17.3.33–17.3.36 of [4]),

$$\textbf{GaussKummer}[h] \;/.\; {}_2F_1(\text{a\_, b\_; c\_; z\_}) \mapsto \text{With}\Big[\{n = c - a - b\},\; \frac{\Gamma(a+b+n)}{\Gamma(a)\,\Gamma(b)}\left(\sum_{k=0}^{\infty} \frac{(a+n)_k\,(b+n)_k}{k!\,(k+n)!}(-\log(1-z) + \psi(k+1) + \psi(k+n+1) - \psi(a+k+n) - \psi(b+k+n))(1-z)^k\right)(z-1)^n + \frac{(n-1)!\,\Gamma(a+b+n)}{\Gamma(a+n)\,\Gamma(b+n)}\sum_{k=0}^{n-1}\frac{(a)_k\,(b)_k\,(1-z)^k}{k!\,(1-n)_k}\Big] \;//\; \text{Simplify}$$

$$\frac{1}{4}\left(\left(\sum_{k=0}^{\infty}\frac{(1-h)^k\left(\frac{3}{2}\right)_k^2\left(-\log(1-h) + \psi^{(0)}(k+1) - 2\psi^{(0)}\left(k+\frac{3}{2}\right) + \psi^{(0)}(k+3)\right)}{k!\,(k+2)!}\right)(h-1)^2 + 4(h+3)\right)$$

Polynomial Approximants

Linear Approximant

From the exact values at $h = 0$,

GaussKummer[0]

π

and at $h = 1$,

GaussKummer[1]

4

one constructs the linear *extreme perfect* approximant.

Linear[h_] = (1 − *h*) GaussKummer[0] + *h* GaussKummer[1] // Simplify

$\pi - h\,(-4+\pi)$

Plot[{GaussKummer[*h*], Linear[*h*]}, {*h*, 0, 1}]



- Graphics -

Quadratic Approximant

The quadratic approximant, exact at $h = 0,\ 1/2,\ 1$,

Table$\left[\{h, \text{GaussKummer}[h]\}, \left\{h, 0, 1, \frac{1}{2}\right\}\right]$ **// FullSimplify**

$$\begin{pmatrix} 0 & \pi \\ \frac{1}{2} & \frac{\sqrt{\frac{\pi}{2}}\,\Gamma\left(\frac{3}{4}\right)}{\Gamma\left(\frac{5}{4}\right)} + \frac{\Gamma\left(\frac{1}{4}\right)\Gamma\left(\frac{5}{4}\right)}{\sqrt{\pi}} \\ 1 & 4 \end{pmatrix}$$

Quadratic[h_] = InterpolatingPolynomial[%, *h*] // *N*

$(0.0891819\,(h-0.5)+0.813816)\,h+3.14159$

has a maximum absolute relative error of $\lesssim 8\times 10^{-4}$.

```
Plot[10^4 (1 - Quadratic[h]/GaussKummer[h]), {h, 0, 1}]
```



- Graphics -

n^{th} –order polynomial Approximant

Here is the n^{th} –order "even–tempered" polynomial approximant, exact at $h = m/n$ for $m = 0, 1, \ldots, n$.

```
poly[n_] := poly[n] = Function[h, Evaluate@InterpolatingPolynomial[
    N@Table[{h, GaussKummer[h]}, {h, 0, 1, 1/n}], h]]
```

The 9^{th} –order approximant has a maximum absolute relative error of $< 10 \times 10^{-6}$.

```
Plot[10^6 (1 - poly[9][h]/GaussKummer[h]), {h, 0, 1}, PlotRange → All, PlotPoints → 30]
```



- Graphics -

Chebyshev polynomial Approximant

Sampling the Gauss–Kummer function at the zeros of $T_n(2x - 1)$, which are at $x_m = \cos^2((m + 1/4)\frac{\pi}{n})$, yields a Chebyshev polynomial approximant.

```
Chebyshevpoly[n_] :=
  Chebyshevpoly[n] = Function[h, Evaluate@InterpolatingPolynomial[
      N@Join[{{0, GaussKummer[0]}, {1, GaussKummer[1]}},
          Table[{cos^2((m + 1/4) π/n), GaussKummer[cos^2((m + 1/4) π/n)]},
           {m, n}]], h]]
```

The 8^{th} –order approximant has a maximum absolute relative error of $\lesssim 7\times 10^{-6}$.

```
Plot[10^6 (1 - Chebyshevpoly[8][h]/GaussKummer[h]), {h, 0, 1}, PlotRange → All]
```



- Graphics -

□ Rational Approximation

After loading the package (stub),

```
<< NumericalMath`
```

one obtains a family of $[N, M]$ rational polynomial minimax approximations.

```
GKapprox[n_, m_] := GKapprox[n, m] = Function[h, Evaluate[
      MiniMaxApproximation[GaussKummer[h], {h, {0, 1}, n, m}][[2, 1]]]]
```

For example, the [4,3] minimax approximation,

```
GKapprox[4, 3][h]
```

$$\frac{-0.0811183\, h^4 + 0.273498\, h^3 + 1.77163\, h^2 - 5.0554\, h + 3.14159}{-0.14146\, h^3 + 1.01321\, h^2 - 1.8592\, h + 1}$$

has (absolute) relative error $\lesssim 2.3\times 10^{-7}$, but is not “extreme perfect”.

$$\text{Plot}\left[10^7\left(1-\frac{\text{GKapprox}[4,3][h]}{\text{GaussKummer}[h]}\right),\{h,0,1\}\right]$$



- Graphics -

Using the linear approximant, $4h+\pi(1-h)$, and noting that $h(1-h)$ vanishes at both $h=0$ and $h=1$, leads to an optimal $[N+2, M]$ extreme perfect approximant of the form

$$\pi\,{}_2F_1\left(-\frac{1}{2},-\frac{1}{2};1;h\right)\approx 4h+\pi(1-h)+\alpha\,h(1-h)\,\frac{\prod_{i=1}^{N}(h-p_i)}{\prod_{j=1}^{M}(h-q_j)},$$

where the parameters α, $\{p_i\}_{i=1,\ldots,N}$, and $\{q_j\}_{j=1,\ldots,M}$ need to be determined. Implementation of the approximant is immediate.

$$\textbf{EllipseApproximant}[\alpha\_,\text{p\_List},\text{q\_List}]:=$$
$$\text{Function}\left[h,\text{Evaluate}\left[4h+\pi(1-h)+\alpha\,h(1-h)\,\frac{\text{Times}\;@@\;(h-p)}{\text{Times}\;@@\;(h-q)}\right]\right]$$

After uniformly sampling the Gauss–Kummer function,

{xdata, ydata} = Table[{*h*, GaussKummer[*h*]}, {*h*, 0, 1, 0.001}] // Transpose;

one can use NMinimize and the ∞–norm to obtain the accurate approximants. For example, the (almost) optimal [3, 2] approximant is computed using

$$\text{NMinimize}\left[\|\text{ydata}-\text{EllipseApproximant}[\alpha,\{p\},\{q,r\}][\text{xdata}]\|_\infty,\begin{pmatrix}\alpha & 0.22 & 0.24\\ p & 1.25 & 1.35\\ q & 3.4 & 3.5\\ r & 1.15 & 1.25\end{pmatrix}\right]$$

$\{0.0000140975, \{p\to 1.28546, q\to 3.475, r\to 1.19671, \alpha\to 0.235456\}\}$

leading to

EllipseApproximant[α, {*p*}, {*q*, *r*}][*h*] /. Last[%]

$$\frac{0.235456\,(h-1.28546)\,h\,(1-h)}{(h-3.475)\,(h-1.19671)}+\pi(1-h)+4h$$

This simple approximant has (absolute) relative error $\lesssim 4\times 10^{-6}$.

$$\text{Plot}\left[10^6\left(1-\frac{\%}{\text{GaussKummer}[h]}\right),\ \{h,0,1\}\right]$$



- Graphics -

■ Conclusions

Mathematica is an ideal tool for developing accurate approximants to special functions because:

- all special functions of mathematical physics are built–in and can be evaluated to arbitrary precision for general complex parameters and variables;
- standard analytical methods—such as symbolic integration, summation, series and asymptotic expansions, and polynomial interpolation—are available;
- properties of special functions—such as identities and transformations—are available at MathWorld [6] and the Wolfram functions Site [7] and, because these properties are expressed in *Mathematica* syntax, can be used directly;
- relevant built–in numerical methods include rational polynomial approximants, minimax methods, and numerical optimization for arbitrary norms;
- visualization of approximants can be used to estimate the quality of approximants; and
- combining these approaches is straightforward and leads, in a natural way, to optimal approximants.

This paper uses the exercise of computing the perimeter of an ellipse using a simple set of approximants to illustrate these points.

■ References

[1] http://www.ebyte.it/library/docs/math05a/EllipsePerimeterApprox05.html

[2] http://www.numericana.com/answer/ellipse.htm

[3] http://www.alumni.iitb.ac.in/mathslecture.htm

[4] Abramowitz and Stegun onine:
http://www.convertit.com/Go/ConvertIt/Reference/AMS55.ASP

[5] Peter Kahlig "A New Elliptic Mean", Sitzungsber. Abt. II (2002) **211** 137-142. URL:
http://hw.oeaw.ac.at/?arp=x−coll7178b/2003−7.pdf.

[6] http://mathworld.wolfram.com/Arithmetic−GeometricMean.html.

[7] http://functions.wolfram.com